Introduction to LanHEP package

- ***Andrei Semenov: V3.0, arXiv:0805.0555***

http://theory.sinp.msu.ru/~semenov/lanhep.html

This is the program for Feynman rules generation in momentum space

- **Example for QED**

$$\mathcal{L}_{QED} = -\frac{1}{4}F_{\mu\nu}F^{\mu\nu} + \bar{e}\gamma^{\mu}(i\partial_{\mu} + g_e A_{\mu})e - m\bar{e}e, \qquad \mathcal{L}_{GF} = -\frac{1}{2}(\partial_{\mu}A^{\mu})^2$$

```
model QED/1.
parameter ge=0.31333:'elementary electric charge'.
spinor e1/E1:(electron, mass me=0.000511).
vector A/A:(photon).
let F^mu^nu=deriv^nu*A^mu-deriv^mu*A^nu.
lterm -1/4*(F^mu^nu)**2 - 1/2*(deriv^mu*A^mu)**2.
lterm E1*(i*gamma*deriv-me)*e1.
lterm ge*E1*gamma*A*e1.
```

Fields in the vertex	Variational derivative of Lagrangian by fields
$E1_a \quad e1_b \quad A_\mu$	$ee \cdot \gamma^{\mu}_{ab}$

Features of LanHEP

- it reads Lagrangian written in the form close to one used in publications and transforms it into momenta space
- it writes Feynman rules in the form of four tables in CompHEP format as well as tables in LaTeX format
- LanHEP expands expression and combines similar terms user can define the substitution rules, it allows to define multiplets, and their components
- user can write Lagrangian terms with Lorentz and multiplet indices explicitly or omit indices (all or some of them)
- LanHEP performs explicit summation over the indices in Lagrangian terms, if the corresponding components for multiplets and matrices are introduced
- it allows user to introduce vertices with 4 fermions and 4 colored particles (such vertices can't be introduced directly in CompHEP) by means of auxiliary field with constant propagator
- it also can check whether the set of introduced vertices satisfies the electric charge conservation law
- many more features: see manual(!) – using superpotential formalism, check for BRST invariance, two-component notation for fermions, ...

QCD as the next example

- **Gauge interactions** $L_{YM} = -\frac{1}{4} F^{a\mu\nu} F^a_{\mu\nu},$

where $F^a_{\mu\nu} = \partial_\mu G^a_\nu - \partial_\nu G^a_\mu - g_s f^{abc} G^b_\mu G^c_\nu,\ G^a_\mu(x)$

- **Quark kinetic term** $L_F = \bar{q}_i \gamma^\mu \partial_\mu q_i + g_s \lambda^a_{ij} \bar{q}_i \gamma^\mu q_j G^c_\mu,$

- **Gauge fixing term and Fadeev-Popov ghost term**

$$\mathcal{L}_{GF} = -\frac{1}{2}(\partial_\mu G^\mu_a)^2 + i g_s f^{abc} \bar{c}^a G^b_\mu \partial^\mu c^c,$$

```
model QCD/2.
parameter gg=1.117:'Strong coupling'.
spinor q/Q:(quark, mass mq=0.01, color c3).
vector G/G:(gluon, color c8, gauge).
let F^mu^nu^a = deriv^nu*G^mu^a - deriv^mu*G^nu^a -
          gg*f_SU3^a^b^c*G^mu^b*G^nu^c.
lterm -F**2/4-(deriv*G)**2/2.
lterm Q*(i*gamma*deriv+mq)*q.
lterm i*gg*f_SU3*ccghost(G)*G*deriv*ghost(G).
lterm gg*Q*gamma*lambda*G*q.
```

QCD Feynman rules generated by LanHEP in LaTeX format

Fields in the vertex	Variational derivative of Lagrangian by fields
$G_{\mu p}\ G.C_q\ G.c_r$	$-gg\cdot p_3^{\mu} f_{pqr}$
$Q_{ap}\ q_{bq}\ G_{\mu r}$	$gg\cdot\gamma_{ab}^{\mu}\lambda_{pq}^{r}$
$G_{\mu p}\ G_{\nu q}\ G_{\rho r}$	$gg f_{pqr}(p_3^{\nu}g^{\mu\rho}-p_2^{\rho}g^{\mu\nu}-p_3^{\mu}g^{\nu\rho}+p_1^{\rho}g^{\mu\nu}+p_2^{\mu}g^{\nu\rho}-p_1^{\nu}g^{\mu\rho})$
$G_{\mu p}\ G_{\nu q}\ G_{\rho r}\ G_{\sigma s}$	$gg^2(g^{\mu\rho}g^{\nu\sigma}f_{pqt}f_{rst}-g^{\mu\sigma}g^{\nu\rho}f_{pqt}f_{rst}+g^{\mu\nu}g^{\rho\sigma}f_{prt}f_{qst}$
	$+g^{\mu\nu}g^{\rho\sigma}f_{pst}f_{qrt}-g^{\mu\sigma}g^{\nu\rho}f_{prt}f_{qst}-g^{\mu\rho}g^{\nu\sigma}f_{pst}f_{qrt})$

Syntax of LanHEP

- ***The LanHEP input file is the sequence of statements, each starts with a special identifier (such as parameter, lterm, etc) and ends with the full-stop '.' symbol. Statement can occupy several lines***
- ***Identifiers:*** ***Indentfiers are the names of particles, parameters etc.***
- ***Constants:*** ***integers, floating point numbers, strings***
- ***Comments:*** '%' ,'/*' ... '*/'
- ***Order of the indices of the objects (default):***

```
[spinor, color c3, color c8, vector]
```

- ***declaring new groups:***

```
group color:SU(3).
repres color:(c3/c3b,c8).
```

- ***parameters***

parameter *name*=*value*:*comment*.

- ***particles***

scalar *P/aP*:(*options*).
spinor *P/aP*:(*options*).
vector *P/aP*:(*options*).

Syntax of LanHEP

- ***Specials*** gamma, gamma5, moment, deriv, lambda, f_SU3
  declaring new specials: `special` *name*`:(`*islist*`).`

- ***Orthogonal matrice*** `OrthMatrix(` $\{\{a_{11}, a_{12}\}, \{a_{21}, a_{22}\}\}$ `).`

- ***Including files*** `read` *file*`.` or `use` *file*`.` (no multiple reading)

- ***Checking electric charge conservation*** `SetEM(`*photon*, *param*`).`

- ***Running LanHEP*** `lhep` *filename options*

  `-OutDir` *directory* Set the directory for output files

  `-InDir` *directory* Set the directory where to search files

  `-tex` LanHEP generates LaTeX files

  `-frc` If `-tex` option is set, forces LanHEP to split 4-fermion and 4-color vertices just as it is made for CompHEP files.

  `-texLines` *num* Set number of lines in LaTeX tables

  `-texLineLength` *num* Controls width of the Lagrangian

LanHEP installation

http://theory.sinp.msu.ru/~semenov/lanhep.html

tar -zxvf lhep*xxx*.tar.gz

cd lhep*xxx*

make

make clean

Exercise#5
install LanHEP

Running LanHEP

- ***../lhep stand.mdl***

File sm_tex processed, 0 sec.

File stand.mdl processed, 1 sec.

user-defined model

$$\bar{b}_{ap} \quad t_{bq} \quad {W^-}_{\mu} \quad \Big| \; -\frac{1}{4}\frac{e\cdot\sqrt{2}\cdot Vtb}{s_w}\cdot(1-\gamma^5)_{cb}\delta_{pq}\gamma^{\mu}_{ac}$$

- ***Let us add left and right anomalous couplings to WtB interaction: Ar and Al***

```
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
%%% anomalous WtB interactions Ar and Al
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

parameter Ar,Al.
let PR=(1+gamma5)/2, PL=(1-gamma5)/2.

lterm   -g/Sqrt2*
         (Ar*anti(t)*'W+'*gamma*PR*b  + Al*anti(t)*'W+'*gamma*PL*b)
        + AddHermConj.
```

$$-\frac{1}{4}\frac{e\cdot\sqrt{2}}{s_w}\delta_{pq}\gamma^{\mu}_{ac}(Vtb\cdot(1-\gamma^5)_{cb}+Ar\cdot(1+\gamma^5)_{cb}+Al\cdot(1-\gamma^5)_{cb})$$

Exercise#6
implement $\sigma^{\mu\nu}$ anomalous terms B_l, B_r

Using the superpotential formalism in the MSSM and its extensions

- ***Superpotential – a polynomial W depending on scalar fields A_i***
- ***The most general form of the MSSM superpotential which does not violate gauge invariance and the SM conservation laws is:***

$$W = \mu\epsilon_{ij}H_i^1H_j^2 + \epsilon_{ij}Y_l^{IJ}H_i^1L_j^IR^J + \epsilon_{ij}Y_d^{IJ}H_i^1Q_j^ID^J + \epsilon_{ij}Y_u^{IJ}H_i^2Q_j^IU^J$$

which in LanHEP notation will take a form

```
keep_lets W.
let W=eps*(mu*H1*H2+ml*H1*L*R+md*H1*Q*D+mu*H2*Q*U).
```

Where H1, H2, L, R, Q, U, D ***should be defined above as doublets and singlets in terms of scalar particles.***

keep_lets ***statement substitution of*** H1, H2, L, R, Q, U, D **in terms of their components**

Using the superpotential formalism in the MSSM and its extensions

- ***Yuakawa interactions are given by***

$$-\frac{1}{2}\left(\frac{\partial^2 W}{\partial A_i \partial A_j}\Psi_i\Psi_j + H.c.\right)$$

which in the LanHEP language will take form

```
lterm - df(W,H1,H2)*fH1*fH2 - ...  + AddHermConj.
```

where fH1, fH2 ***should be defined above as fermionic partners of corresponding multiples, e.g.***

```
let f_h1 = { Zn31*up(~o1)+Zn32*up(~o2)+Zn33*up(~o3)+Zn34*up(~o4),
             Zm21*up('~1-')+Zm22*up('~2-') }.
```

Using the superpotential formalism in the MSSM and its extensions

- ***FF^* term from scalar supersymmetric potential***

$$V = \frac{1}{2} D^a D^a + F_i^\star F_i \quad \textit{where} \quad F_i = \partial W / \partial A_i$$

in LanHEP notation will take a form

```
lterm - df(W,H1)*df(Wc,H1c) - ....
```

where Wc ***should be decleared above as the conjugate superpotential***

FF^* term can be introduced even in shorter way as

```
lterm - dfdfc(W,H1) - ....
```

where dfdfc(W,H1) **function** ***evaluates the variational derivative, multiplies it by the conjugate expression and returns the result***

Vertices with color particles in CalcHEP

- *The CalcHEP Lagrangian tables do not describe explicitly the color structure of a vertex.*
- *If color particles are present in the vertex, the following implicit contractions are assumed (p, q, r are color indices):*
  - δ_{pq} for two color particles $A^{\bar{r}}_{p}$ and A^{r}_{q}
  - λ^{r}_{pq} for three particles, which are color triplet, antitriplet and octet
  - f^{pqr} for three color octets $f^{pqr} G^{p}_{\mu} G^{q}_{\nu} G^{r}_{\lambda}$
  - ***There are no other color structures in CalcHEP***

Vertices with color particles in CalcHEP

- *4-gluon vertex can be split it into 3-legs vertices* $f^{pqr} G^q_\mu G^r_\nu X^p_{\mu\nu}$



- *Here the field* $X^p_{\mu\nu}$ *is a Lorenz tensor and color octet, and this field has constant propagator.*
- *If gluon name in CalcHEP is 'G', the name 'G.t' is used for this tensor particle; its indices are denoted as 'm_' and 'M_' ('_' is the number of the particle in table item).*

Vertices

Clr	Del	Size	Read	ErrMes			
A1	|A2	|A3	|A4	|>	Factor	<|>	Lorentz part
G	|G	|G	|	|GG			|m1.m2*(p1-p2).m3+m2.m3*(p2-p3).m1+m3.m1*(p3-p1).m2
G	|G	|G.t	|	|GG/Sqrt2			|m1.M3*m2.m3-m1.m3*m2.M3

Vertices with color particles in LanHEP

- *The splitting of vertex with 4 colored particle into 3-particles vertices is done by LanHEP automatically: each vertex containing 4 color particles is split to 2 vertices which are joined by automatically generated auxiliary field*
- *option SplitCol1=N.*
  where N is a number:
  - *-1 remove all vertices with 4 color particles from Lagrangian;*
  - *0 turn off multiplet level vertices splitting;*
  - *1 allows vertices splitting with 4 color multiplets;*
  - *2 allows vertices splitting with any 4 scalar multiplets except Higgs*
- *option SplitCol2=N.*
  - *where N is a number:*
  - *0 disable vertex level splitting;*
  - *1 enable vertex level splitting (only for vertices with 4 color particles).*
- *the default value is 2 for SplitCol1 and 1 for SplitCol2*